

Economic equilibrium and optimization problems using GAMS

Notes 1: Optimization

James R. Markusen
University of Colorado Boulder

Simple optimization problem:

Competitive firm produces X and can sell at a fixed price p

Cost is given by

$$C(X) = aX + (b/2)X^2 + FC \quad \text{where } FC \text{ is a fixed cost}$$

Objective: maximize profit. Profit is given by

$$\Pi = \text{revenues} - \text{costs} = pX - aX - (b/2)X^2 - FC$$

The intuitive, heuristic approach is: start with a small value X and calculating profits. Increase X a little and if profits increase, then increase X a little more.

As long as an increase in X increases profits, keep increasing X . Stop when profits no longer increase.


This is the intuition behind the **first-order condition** for profit maximization: find the X at which the **first derivative** of profit with respect to (wrt) X is zero.

$$\frac{d\pi}{dX} = p - (a + bX) = 0 \quad \text{FOC} \quad X^* = \frac{p - a}{b}$$

Note that $a + bX$ is **marginal cost**, so this condition is often written as $p = mc$, where $mc = a + bX$.

However, if we just look at this equation and not the path, this could be a minimum, not a maximum.

Illustration of first & second order, entry condition



Second-order condition is that profits are decreasing on either side of this (proposed) optimal value. This implies a restriction on the **second derivative**, the **second-order condition**.

$$\frac{d^2\pi}{dX^2} = -b < 0 \quad \text{SOC}$$


However, and this is rarely mentioned in economics or math books, the existence of fixed costs adds a complication.

If those costs are not already paid (referred to as “sunk costs”), then there is an **entry condition**:

$$\pi \geq 0 \quad \text{at} \quad X = X^* \quad \text{Entry condition}$$

This is *revenue* \geq *total cost*. Dividing through both sides this is often expressed as *price* \geq *average cost*.

Cost curves and profit for simple maximization problem in Notes 0 - optimization



Parameter values:

$$A = 2$$

$$B = 1$$

$$P = 4$$

Profit maximization \Leftrightarrow price = marginal cost

Not quite done yet! “Corner solutions” and complementarity.

Suppose that the FC is zero or “sunk”: already paid, not recoverable.

It could be the case that price is so low that it is less than marginal cost even at zero output.

Economic variables such as prices and quantities are inherently bounded as non-negative: you cannot run production in reverse.

This is then the case of a corner solution, meaning that the optimal output is zero.

The correct first-order condition should actually be

$$\frac{d\pi}{dX} = p - (a + bX) = p - mc \leq 0 \quad \perp \quad X \geq 0$$

where the symbol \perp means “complementary to”.

Complementarity means that if one of these conditions is a strict inequality then the other one is a strict equality.

if the first-order condition holds as a strict inequality, the complementary non-negative variable is zero.

if the first-order condition holds as a strict equality, the complementary non-negative variables is positive.

$$[p - (a + bX)]X = 0$$

Karush-Kuhn-Tucker theorem and conditions (KKT)

Complementarity is formalized in the KKT theorem, which gives necessary conditions for a solution to an optimization problem.

Suppose that we want to maximize profits, subject to X being non-negative.

$$\text{Max } \pi = pX - (aX + (b/2)X^2) - FC \quad \text{s.t. } X \geq 0$$


KKT: there exists a non-negative number σ such the necessary conditions for an optimum are given by:

$$p - (a + bX) + \sigma = 0 \quad \sigma X = 0$$

σ is often referred to as a “**slack variable**”; in GAMS it is the “**marginal value**” and is calculated and displayed as part of the solution. KKT conditions are sometimes called “complementary slackness”.

The important thing about KKT is that it proves that an **optimization problem** can be **converted into a set of equations**, which are much easier to solve than weak inequalities.

Illustration of corner solution and interpretation of slack variable



Value functions and the envelop theorem.

$$\pi = pX - aX - (b/2)X^2 \quad \Rightarrow \quad X = \frac{p - a}{b}$$

$$\pi = (p - a)X - (b/2)X^2 = (p - a)\frac{(p - a)}{b} - \frac{b}{2}\frac{(p - a)^2}{b^2}$$

$$\pi(p) = \frac{(p - a)^2}{2b} \quad \text{is a value function: the maximized value of profits for given } p, a, \text{ and } b$$

There is something very cool about value functions

$$\frac{d\pi}{dp} = \frac{(p - a)}{b} = X^* = \text{the optimal output level.}$$

This important and valuable result is a consequence of the **envelop theorem**.

When profits are maximized, we derive the optimal amount of X to produce at every level of the exogenous variable p (price out output): $X = X(p)$, read “ X is a function of p ”.

Ok, how are profits affected by an (exogenous) change in p ?

$$\pi = pX(p) - aX(p) - (b/2)X(p)^2$$

Differentiate π with respect to p (sometimes referred to as “comparative statics”: note p appears twice in first term, requires differentiation of a product rule.

$$\frac{d\pi}{dp} = \left[\frac{\partial \pi}{\partial p} \right]_{\bar{X}} + \left[\frac{\partial \pi}{\partial X} \right] \frac{dX}{dp} = X + [p - a - bX] \frac{dX}{dp}$$

Since X is already chosen optimally for the initial p ,

$$\frac{\partial \pi}{\partial X} = [p - a - bX] = 0$$


At the optimal X , the term in square brackets, is $[p - mc] = 0$

$$\frac{d\pi}{dp} = X = \left[\frac{\partial \pi}{\partial p} \right]_{\bar{X}}$$

The intuition for “envelop” is provided by the attached figure.

This will prove extremely useful for us, especially when applied to cost functions: the derivative of cost with respect to a factor prices (e.g., the wage rate) is the optimal amount of the factor (e.g., labor) to use at current factor prices.

Illustration of the Envelop Theorem for profit maximization



Coding an economic equilibrium problem in GAMS

Title (not needed), comment statements can be used at the beginning of the code, preceded with a *, in the *first column* of a line. Or using a block: \$ontext - \$offtext

```
$TITLE: M1, model illustrating nlp, mcp and mpec solvers
```

```
$ONTEXT
```

```
competitive firm maximizes profits
```

```
X is the output produced from some inputs (not specified)
```

```
P is the price of output
```

```
 $C(X) = aX + (b/2)*X$  is the cost function.
```

```
 $MC(X) = a + b*X$  is marginal cost (derivative of  $C(X)$  wrt  $X$ )
```

```
objective of firm is to maximize profits
```

```
PROFIT =  $P*X - C(X) = P*X - (aX + (b/2)*X**2)$ 
```

```
$OFFTEXT
```

Begin a series of declaration and assignment statements.

PARAMETERS

```
P      price of good X
A      linear parameter in cost function
B      quadratic parameter in cost founction;
```

Parameters must be assigned values before the model is solved

```
A = 2; B = 1; P = 4;
```

Declare a list of variables. Any variable to be maximized or minimized must be unbounded, and called a VARIABLE.

Other economic variables such as prices and quantities are naturally restricted to be non-negative, so declaring them as “nonnegative variables” tells GAMS to set lower bounds of zero.

VARIABLES

```
PROFIT  profits - to be maximized;
```

NONNEGATIVE VARIABLES

```
X          firm output;
```

Declare equations.

EQUATIONS

```
EPROFIT  equation for profit
```

```
FOC      first order condition for profit maximization;
```

Specify equations. Format: equation name followed by two periods

The equation is written with =E= (for any objective function)

For weak inequalities, =G= for “greater than or equal to”.

```
EPROFIT..  PROFIT =E= P*X - (A*X + (B/2)*X**2);
```

```
* set initial starting values using .L (level) notation - important!
```

```
X.L = 1;
```


Next model definition and solve statement.

```
* solved directly using NLP
MODEL PMAXNLP /EPROFIT/;
SOLVE PMAXNLP USING NLP MAXIMIZING PROFIT;
```

First-order condition for solving the problem as an mcp.

```
FOC..      A + B*X =G= P;
```

For an mcp (constructs the underlying KKK conditions), a model declaration must have matched equations (weak inequalities) and unknowns. The syntax is <equation name>.<variable name>

```
* solved as an MCP using the first-order (KKT) condition
MODEL PMAXMCP /FOC.X/;
SOLVE PMAXMCP USING MCP;
```

Extracting and displaying results, performing counter-factuals. First declare some additional parameters

PARAMETERS

```
XOPT0    optimal X output
PROPT0    optimal profit
XOPT1    optimal X output under counterfactual P = 8
PROPT1    optimal profit under counterfactual P = 8
XOPT2    X output under government revenue maximizing tax
PROPT2    profit under government revenue maximizing tax
DWTLOSS   deadweight loss from the tax;
```

To set a parameter at the solution value of a variable, use the .L notation (L for level, because GAMS also stores other values of a variable: lower bound, upper bound, and marginal).

```
XOPT0    = X.L;
PROPT0    = P*X.L - (A*X.L + (B/2)*X.L**2);
```

GAMS does not automatically print parameters values

```
DISPLAY XOPT0 , PROPT0 ;
```

Counter-factuals (scenarios) are done by simply changing parameter values. There is no need to re-declare the model if there are no other changes.

```
* counter-factual experiments (scenarios) with the model  
* raise price to P = 8 ;
```

```
P = 8 ;
```

```
SOLVE PMAXMCP USING MCP ;
```

```
XOPT1 = X.L ;
```

```
PROPT1 = P*X.L - (A*X.L + (B/2)*X.L**2) ;
```

```
DISPLAY XOPT0 , XOPT1 , XOPT1 , PROPT1 ;
```

Part II: adding constraint equations, **Lagrangean formulation** of KKT conditions

Suppose we take our previous model and add the complication that there is a capacity constraint on X , a maximum amount of output that can be produced.

Let $XCAP$ denote the maximum output, so that there is a constraint on the optimization problem: $XCAP \geq X$.

There is a method for deriving the KKT conditions for this problem that in economics we refer to as a “Lagrangean function”.

This adds in the constraint to the objective function with the constraint multiplied by a **slack variable** or **Lagrangean multiplier**.

Lagrangian function for capacity constrained problem (we'll ignore the slack non-negativity conditions for clarity here, ignore FC).

$$\text{Max } \pi = pX - (aX + (b/2)X^2) + \sigma(XCAP - X)$$

This is an equation in two unknowns. The FOCs for the problem and the complementarity relationships are:

$$\frac{\partial \pi}{\partial X} = p - (a + bX) - \sigma \leq 0 \quad \perp \quad X \geq 0$$

$$\frac{\partial \pi}{\partial \sigma} = XCAP - X \geq 0 \quad \perp \quad \sigma \geq 0$$

Note that the second condition implies $\sigma(XCAP - X) = 0$. If the capacity constraint is not binding at the solution, the slack variable (Lagrangian multiplier) is zero and then from the first condition, price equals marginal cost.

Capacity constraint not binding:

$$p = mc = a + bX, \quad XCAP \geq X, \quad \sigma = 0$$

Capacity constraint binding implies price > marginal cost

$$p = a + bX + \sigma > mc, \quad XCAP = X, \quad \sigma > 0$$

The slack variable gives the “cost” of the capacity constraint: the difference between price and marginal cost of production.

Alternatively, the slack variable gives the increase in profits from loosening the constraint by one unit. This is another Envelop Theorem result: the term in brackets is the FOC = 0.

$$\frac{\partial \pi}{\partial XCAP} = \sigma + [p - (a + bX) - \sigma] \frac{dX}{dXCAP} = \sigma$$

Illustration of constrained optimization and slack variable
(slack variables = Lagrangean multiplier)

